Álgebra Lineal y Estructuras Matemáticas

J. C. Rosales y P. A. García Sánchez

Departamento de Álgebra, Universidad de Granada

Capítulo 1

Conjuntos, relaciones y aplicaciones

1. Conjuntos

La idea de conjunto es una de las más significativas en Matemáticas. La mayor parte de los conceptos matemáticos están construidos a partir de conjuntos. (Existe una aproximación funcional basada en el λ-cálculo y la Lógica Combinatoria, que hoy en día han tenido una papel fundamental en la programación funcional.)

Podríamos decir que un conjunto es simplemente una colección de objetos a los que llamaremos elementos del conjunto. Esta definición nos bastará para los contenidos de este curso, pero desde el punto de vista matemático es imprecisa y da lugar rápidamente a paradojas. Desde comienzos del siglo XX esta definición dejó de utilizarse por los problemas que acarrea. Por desgracia, dar una definición precisa está bastante lejos de los objetivos de este guión.

- Cuando x sea un elemento de un conjunto A, escribiremos $x \in A$, que se lee "x pertenece a A".
- Diremos que un conjunto A es subconjunto del conjunto B, y lo denotaremos por $A \subseteq B$, si todo elemento de A pertenece a B.
- Un conjunto A es igual que otro conjunto B si tienen los mismos elementos, a saber, si $A \subseteq B$ y $B \subseteq A$. Cuando esto ocurre, escribiremos $A = B$.
- Admitiremos la existencia de un conjunto sin elementos, al que denotemos por $\emptyset$ y llamaremos conjunto vacío. El conjunto vacío es subconjunto de cualquier conjunto.

2. Operaciones con conjuntos

Sean A y B conjuntos.

1) La intersección de A y B es el conjunto formado por los elementos comunes de A y de B, y lo denotamos así
$$A \cap B = \{x \text{ tales que } x \in A \text{ y } x \in B\}.$$
2) La unión de A y B es el conjunto formado al tomar todos los elementos de A y los de B.
$$A \cup B = \{x \text{ tales que } x \in A \text{ o } x \in B\}.$$
3) La diferencia de A y B es el conjunto que tiene por elementos los elementos de A que no están en B.
$$A \setminus B = \{x \in A \text{ tales que } x \notin B\}$$
(siempre que tachemos un símbolo, estamos indicando que no se cumple la condición sin tachar; así $x \notin B$ significa que x no pertenece a B, $A \neq B$ significa que A es distinto de B, etcétera).
4) $\mathcal{P}(A) = \{X \text{ tales que } X \subseteq A\}$ es el conjunto de partes de A o conjunto potencia de A.
5) El producto cartesiano de A y B es el conjunto de parejas cuya primera componente está en A y la segunda en B. Esto se escribe de la siguiente forma.
$$A \times B = \{(a, b) \text{ tales que } a \in A \text{ y } b \in B\}.$$
Si en vez de dos conjuntos tenemos $A_1, \ldots, A_n$,
$$A_1 \times \cdots \times A_n = \{(a_1, \ldots, a_n) \text{ tales que } a_1 \in A_1, \ldots, a_n \in A_n\},$$

y a los elementos de $A_1 \times \cdots \times A_n$ les llamaremos n-uplas.

Al conjunto $A \times \overset{n}{\cdots} \times A$ lo denotaremos por A^n, para n un entero positivo.

El cardinal de un conjunto es el número de elementos que contiene. Usaremos $\sharp A$ para denotar el cardinal del conjunto A.

- $\sharp\mathcal{P}(A) = 2^{\sharp A}$.
- $\sharp(A \times B) = \sharp A \cdot \sharp B$.

Maxima 1: Los conjuntos en `maxima` se pueden definir usando llaves o bien la función `set`.

```
(%i1) {a,a,b,c};
```

(%o1) $\{a, b, c\}$

Definamos un par de conjuntos y veamos cómo se pueden hacer las operaciones hasta ahora descritas con ellos.

```
(%i2) A:{1,2,3,4};
```

(%o2) $\{1, 2, 3, 4\}$

```
(%i3) B:set(3,4,5);
```

(%o3) $\{3, 4, 5\}$

```
(%i4) elementp(5,A);
```

(%o4) $false$

```
(%i5) elementp(1,A);
```

(%o5) $true$

```
(%i6) is (A=B);
```

(%o6) $false$

```
(%i7) is (A=A);
```

(%o7) $true$

```
(%i8) setequalp(A,B);
```

(%o8) $false$

```
(%i9) subsetp(A,B);
```

(%o9) $false$

```
(%i10) subsetp(A,union(A,B));
```

(%o10) $true$

```
(%i11) intersection(A,B);
```

(%o11) $\{3, 4\}$

```
(%i12) union(A,B);
```

(%o12) $\{1, 2, 3, 4, 5\}$

```
(%i13) setdifference(A,B);
```

(%o13) $\{1, 2\}$

```
(%i14) powerset(B);
```

(%o14) $\{\{\},\{3\},\{3,4\},\{3,4,5\},\{3,5\},\{4\},\{4,5\},\{5\}\}$

Nótese que el conjunto vacío se denota por {}.

```
(%i15) is(cardinality(powerset(A))=2^(cardinality(A)));
```

(%o15) $true$

```
(%i16) cartesian_product(A,B);
```

(%o16) $\{[1,3],[1,4],[1,5],[2,3],[2,4],[2,5],[3,3],[3,4],[3,5],[4,3],[4,4],[4,5]\}$

Podemos además elegir los elementos de A que son impares.

```
(%i17) subset(A,oddp);
```

(%o17) $\{1,3\}$

O bien las sumas de los pares del producto cartesiano con A y B.

```
(%i18) makeset(a+b, [a,b], cartesian_product(A,B));
```

(%o18) $\{4,5,6,7,8,9\}$

Maxima 2: Pongamos un ejemplo de una función cuyos argumentos sean conjuntos. Podemos definir la diferencia simétrica de dos conjuntos A y B como $(A \setminus B) \cup (B \setminus A)$.

```
(%i1) A:{1,2,3,4};
```

(%o1) $\{1,2,3,4\}$

```
(%i2) B:set(3,4,5);
```

(%o2) $\{3,4,5\}$

```
(%i3) dif_sim(X,Y):=union(setdifference(X,Y),setdifference(Y,X))$
```

Para definir funciones usamos := en vez de : . El "$" al final de una línea inhibe la salida.

```
(%i4) dif_sim(A,B);
```

(%o4) $\{1,2,5\}$

Maxima 3: Podemos definir conjuntos utilizando listas y viceversa, lo cual hace que podamos usar las funciones específicas para listas en conjuntos. Además se pueden definir subconjuntos utilizando funciones booleanas, tal y como vemos a continuación.

```
(%i1) l:makelist(i,i,1,100)$ A:setify(l)$
```

Crea un conjunto con los los enteros del uno al cien.

```
(%i3) B:subset(A,primep);
```

(%o3) $\{2,3,5,7,11,13,17,19,23,29,31,37,41,43,47,53,59,61,67,71,73,79,83,89,97\}$

Escojo aquellos que son primos.

```
(%i4) C:subset(B,lambda([x],is(x>80)));
```

(%o4) $\{83, 89, 97\}$

De entre ellos me quedo con los mayores de 80, que equivale a hacer lo siguiente (ahorrándome la definición de `f`, usando para ello `lambda`, que define de forma anónima una función).

```
(%i5) f(x):=is(x>80)$
(%i6) D:subset(B,f);
```

(%o6) $\{83, 89, 97\}$

3. Relaciones de equivalencia

Sea A un conjunto. Una relación binaria en A es un subconjunto R de $A \times A$. Cuando $(x, y) \in R$ escribimos $x\ R\ y$ y decimos que x está relacionado (mediante R) con y.

Una relación binaria R sobre un conjunto A es una relación de equivalencia si verifica las siguientes propiedades.

1) Para todo $a \in A$, $a\ R\ a$ (R es reflexiva).
2) Dados $a, b \in A$, si $a\ R\ b$, entonces $b\ R\ a$ (R es simétrica).
3) Para cualesquiera $a, b, c \in A$, si $a\ R\ b$ y $b\ R\ c$, entonces $a\ R\ c$ (R es transitiva).

Si R es una relación de equivalencia sobre un conjunto A, y a es un elemento de A, entonces la clase de a es el conjunto de todos los elementos de A que están relacionados con a,

$$[a] = \{x \in A \text{ tales que } x\ R\ a\}.$$

Se define el conjunto cociente de A por R como el conjunto de todas las clases de equivalencia de elementos de A, y se denota por A/R. Así

$$\frac{A}{R} = \{[a] \text{ tales que } a \in A\}.$$

Para una relación de equivalencia R en un conjunto A se tiene que

1) $a\ R\ b$ si y sólo si $[a] = [b]$,
2) $a \not R\ b$ si y sólo si $[a] \cap [b] = \emptyset$.

Ejercicio 1: En el conjunto $\mathbb{Z} = \{0, 1, -1, 2, -2, \ldots\}$ de los números enteros, definimos la siguiente relación de equivalencia.

$$x\ R\ y \text{ si } x - y \text{ es múltiplo de } 5.$$

a) Demuestra que R es una relación de equivalencia.
b) Calcula $[2]$.
c) Describe el conjunto cociente $\frac{\mathbb{Z}}{R}$.
d) ¿Qué cardinal tiene $\frac{\mathbb{Z}}{R}$?

Ejercicio 2: En el conjunto $\mathcal{P}(\{1, 2, 3\})$, definimos la siguiente relación binaria.

$$A \sim B \text{ si } \#A = \#B.$$

a) Demuestra que $\sim$ es una relación de equivalencia.
b) Calcula $[\{1, 2\}]$.
c) Describe el conjunto cociente $\frac{\mathcal{P}(\{1,2,3\})}{\sim}$.
d) ¿Cuántos elementos tiene dicho conjunto cociente?

Dado un conjunto X, una partición de X es una familia de subconjuntos de X, $\{A_i\}_{i \in I}$ $(= \{A_i \text{ tales que } i \in I\})$, de forma que

1) $A_i \neq \emptyset$ para todo $i \in I$,
2) $A_i \cap A_j = \emptyset$ para cualesquiera $i, j \in I$ con $i \neq j$,
3) $X = \bigcup_{i \in I} A_i$ (la unión de todos los elementos de la familia $\{A_i\}_{i \in I}$).

- Se puede comprobar fácilmente que el hecho de ser R una relación de equivalencia sobre A hace que A/R sea una partición de A.
- Es más, si $\{A_1, \ldots, A_n\}$ es una partición de A, entonces

$$R = (A_1 \times A_1) \cup \cdots \cup (A_n \times A_n)$$

es una relación de equivalencia sobre A (nótese que para $a, b \in A$, $a \mathrel{R} b$ si y sólo si existe $i \in \{1, \ldots, n\}$ tal que $a, b \in A_i$) y

$$\frac{A}{R} = \{A_1, \ldots, A_n\}.$$

`Maxima 4:` Veamos cómo se pueden calcular las clases de equivalencia del conjunto $A = \{1, \ldots, 10\}$ sobre la relación de equivalencia $x \mathrel{R} y$ si $x - y$ es un múltiplo de 3.

Primero definimos el conjunto $\{1, \ldots, 10\}$. Para ello hacemos una lista con los elementos del uno al diez, y luego la convertimos en conjunto.

```
(%i1) l:makelist(i,i,1,10);
```

(%o1) $$[1, 2, 3, 4, 5, 6, 7, 8, 9, 10]$$

```
(%i2) s:setify(l);
```

(%o2) $$\{1, 2, 3, 4, 5, 6, 7, 8, 9, 10\}$$

```
(%i3) equiv_classes(s,lambda([x,y],is(remainder(x-y,3)=0)));
```

(%o3) $$\{\{1, 4, 7, 10\}, \{2, 5, 8\}, \{3, 6, 9\}\}$$

También podríamos haber definido R, y luego calculado A/R.

```
(%i4) R(x,y):=is(remainder(x-y,3)=0);
```

(%o4) $$R(x, y) := \mathrm{is}\,(\mathrm{remainder}\,(x - y, 3) = 0)$$

```
(%i5) equiv_classes(A,R);
```

(%o5) $$\{\{1, 4, 7, 10\}, \{2, 5, 8\}, \{3, 6, 9\}\}$$

Se ve que es una partición de A, pues todos sus elementos son no vacíos, disjuntos dos a dos, y la unión de ellos da A.

4. Relaciones de orden

Una relación binaria $\leq$ sobre un conjunto A es una relación de orden si verifica las siguientes propiedades.

1) Para todo $a \in A$, $a \leq a$ (reflexiva).
2) Dados $a, b \in A$, si $a \leq b$ y $b \leq a$, entonces $a = b$ (antisimétrica).
3) Para cualesquiera $a, b, c \in A$, si $a \leq b$ y $b \leq c$, entonces $a \leq c$ (transitiva).

Ejemplos de orden son $\leq$ en $\mathbb{N}$, $\mathbb{Z}$, $\mathbb{Q}$ y $\mathbb{R}$.

Si un conjunto A tiene una relación de orden $\leq$, al par $(A, \leq)$ lo llamaremos conjunto ordenado.

Ejercicio 3: En el conjunto de los números naturales $\mathbb{N} = \{0, 1, 2, \ldots\}$ definimos la relación $a \mid b$ si b es múltiplo de a. Demuestra que $\mid$ es una relación de orden.

Ejercicio 4: Sea X un conjunto. Demuestra que $\subseteq$ es una relación de orden en $\mathcal{P}(X)$.

Un conjunto ordenado $(A, \leq)$ es totalmente ordenado si para cada $a, b \in A$, se tiene que $a \leq b$ o $b \leq A$.

Ejercicio 5: En $\mathbb{N}^n$ definimos la siguiente relación binaria

$$(a_1, \ldots, a_n) \leq_p (b_1, \ldots, b_n) \text{ si } a_1 \leq b_1, \ldots, a_n \leq b_n.$$

Demuestra que $\leq_p$ es una relación de orden (orden producto cartesiano), pero no es un orden total para $n \geq 2$.

Ejercicio 6: En $\mathbb{N}^n$ definimos la siguiente relación binaria $(a_1, \ldots, a_n) \preceq_{\text{lex}} (b_1, \ldots, b_n)$ si la primera coordenada no nula de $(a_1 - b_1, \ldots, a_n - b_n) \in \mathbb{Z}^n$ es positiva (caso de que exista, es decir, puede ser que todas sean nulas). Demuestra que $\preceq_{\text{lex}}$ es un orden total.

4.1. Elementos notables de un conjunto ordenado. Sea $(A, \leq)$ un conjunto ordenado y sea B un subconjunto de A.

1) Decimos que m es un elemento maximal de B si $m \in B$ y para cualquier $b \in B$ tal que $m \leq b$ se tiene que $m = b$.
2) Decimos que m es un elemento minimal de B si $m \in B$ y para cualquier $b \in B$ tal que $b \leq m$ se tiene que $m = b$.
3) Un elemento $m \in B$ es el máximo de B si $b \leq m$ para todo $b \in B$.
4) Un elemento $m \in B$ es el mínimo de B si $m \leq b$ para todo $b \in B$.
5) Decimos que $c \in A$ es una cota inferior de B si $c \leq b$ para todo $b \in B$.
6) Decimos que $c \in A$ es una cota superior de B si $b \leq c$ para todo $b \in B$.
7) Un elemento $s \in A$ es el supremo de B si es el mínimo de todas las cotas superiores de B.
8) Un elemento $i \in A$ es el ínfimo de B si es el máximo de todas las cotas inferiores de B.

Ejercicio 7: En $(\mathbb{N}, |)$, calcula los elementos notables de $\{1, 2, 3, 4, 5\}$.

Maxima 5:

```
(%i1) menores(x,rel,conj):=subset(conj,lambda([y],rel(y,x) ))$
(%i2) mayores(x,rel,conj):=subset(conj,lambda([y],rel(x,y) ))$
(%i3) D:setdifference(divisors(30),{1,2,30});
```

(%o3) $\{3, 5, 6, 10, 15\}$

```
(%i4) menores(15,lambda([x,y],is(mod(y,x)=0)), {1,2,3,4,5,6,7});
```

(%o4) $\{1, 3, 5\}$

```
(%i5) minimal(x,rel,con):=is(menores(x,rel,con)={x}) and elementp(x,con)$
(%i6) maximal(x,rel,con):=is(mayores(x,rel,con)={x}) and elementp(x,con)$
(%i7) minimal(3,lambda([x,y],is(mod(y,x)=0)), D);
```

(%o7) true

```
(%i8) minimales(rel,con):=subset(con,lambda([x],minimal(x,rel,con)))$
(%i9) maximales(rel,con):=subset(con,lambda([x],maximal(x,rel,con)))$
```

```
(%i10) div(x,y):=is(mod(y,x)=0)$
(%i11) minimales(div,D);
```
(%o11) $\{3,5\}$

```
(%i12) maximales(div,D);
```
(%o12) $\{6,10,15\}$

```
(%i13) minimo(rel,con):=block(local(m),
        m:listify(minimales(rel,con)),
       if (is(length(m)=1)) then m[1] else
       error ("Error no hay minimo"))$
(%i14) maximo(rel,con):=block(local(m),
       m:listify(maximales(rel,con)),
       if (is(length(m)=1)) then m[1] else
       error("Error no hay maximo"))$
(%i15) maximo(div,D);
```
Error no hay maximo
#0: maximo(rel=div,con=3,5,6,10,15) – an error. To debug this try: debugmode(true);

```
(%i16) minimo(div,D);
```
Error no hay minimo
#0: minimo(rel=div,con=3,5,6,10,15) – an error. To debug this try: debugmode(true);

```
(%i17) cotasuperior(x,rel,con):=is(con=menores(x,rel,con))$
(%i18) cotainferior(x,rel,con):=is(con=mayores(x,rel,con))$
(%i19) cotainferior(1,div,D);
```
(%o19) true

```
(%i20) cotassuperiores(rel,con,amb):=subset(amb,lambda([x],cotasuperior(x,rel,con)))$
(%i21) cotasinferiores(rel,con,amb):=subset(amb,lambda([x],cotainferior(x,rel,con)))$
(%i22) cotasinferiores(div,D,divisors(30));
```
(%o22) $\{1\}$

```
(%i23) cotasinferiores(div,D,D);
```
(%o23) $\{\}$

```
(%i24) supremo(rel,con,amb):=minimo(rel,cotassuperiores(rel,con,amb))$
(%i25) infimo(rel,con,amb):=maximo(rel,cotasinferiores(rel,con,amb))$
(%i26) supremo(div,D,D);
```
Error no hay minimo
#0: maximo(rel=div,con=)
#1: supremo(rel=div,con=3,5,6,10,15,amb=3,5,6,10,15) – an error. To debug this try: debugmode(true);

```
(%i27) infimo(div,D,divisors(30));
```
(%o27) 1

```
(%i28) supremo(div,D,divisors(30));
```
(%o28) 30

5. Aplicaciones entre conjuntos

Sean A y B dos conjuntos. Una aplicación f de A en B, que denotaremos como $f : A \to B$, es una correspondencia que a cada elemento de A le asocia un único elemento de B (de nuevo esta definición es algo imprecisa, pero suficiente para nuestro curso). Si $a \in A$, al elemento que le asocia f en B lo denotamos por $f(a)$, y se llama la imagen de a por f. Los conjuntos A y B son el dominio y codominio de f, respectivamente. Llamaremos conjunto imagen de f a

$$\mathrm{Im}(f) = \{f(a) \text{ tales que } a \in A\}.$$

Ejercicio 8: Sea $\mathbb{Q}$ el conjunto de los números racionales y $\mathbb{R}$ el de los reales. ¿Tiene sentido decir que $f : \mathbb{Q} \to \mathbb{R}, x \mapsto \frac{x+1}{x-1}$ es una aplicación?

Ejercicio 9: Dada la aplicación $f : \mathbb{N} \to \mathbb{Z},\ f(n) = 2n + 1$. Calcula $\mathrm{Im}(f)$.

5.1. Tipos especiales de aplicaciones.

Si $f : A \to B$ es una aplicación, diremos que f es

1) inyectiva si $f(a) = f(a')$ para $a, a' \in A$, implica $a = a'$;
2) sobreyectiva si $\mathrm{Im}(f) = B$ (para todo $b \in B$, existe $a \in A$ tal que $f(a) = b$);
3) biyectiva si es inyectiva y sobreyectiva.

Ejercicio 10: Demuestra que la aplicación $f : \mathbb{Q} \to \mathbb{R}$ definida por $f(x) = \frac{1}{2}(2x + 1)$ es inyectiva pero no sobreyectiva.

Ejercicio 11: Demuestra que la aplicación $f : \mathbb{Z} \to \mathbb{N},\ f(x) = |x|$ (valor absoluto) es sobreyectiva pero no inyectiva.

Ejercicio 12: Demuestra que la aplicación $f : \mathbb{Q} \to \mathbb{Q}$, $f(x) = \frac{3x+1}{2}$ es biyectiva.

5.2. Composición de aplicaciones.

Sean $f : A \to B$ y $g : B \to C$ dos aplicaciones. La aplicación composición de f y g (también conocida como f compuesta con g) es la aplicación $g \circ f : A \to C$, definida como $(g \circ f)(a) = g(f(a))$. Para calcular la imagen de un elemento por la composición primero aplicamos f y luego g.
Ejercicio 13: Sean $f : \mathbb{Z} \to \mathbb{Z}$, $x \mapsto x^2$, y $g : \mathbb{Z} \to \mathbb{Q}$, $y \mapsto \frac{1}{2}(y + 1)$. Calcula $g \circ f$.

- La composición de aplicaciones es asociativa ($f\circ(g\circ h) = (f\circ g)\circ h$) pero no es conmutativa ($f \circ g$ no tiene por qué ser igual a $g \circ f$).

Maxima 6: Veamos como las funciones cuadrado y sumar uno no conmutan al componerlas.

```
(%i1) f(x):=x^2$ g(x):=x+1$
(%i2) f(g(1)); g(f(1));
```

(%o2) 4

(%o3) 2

```
(%i4) f(g(x))=g(f(x));
```

(%o4) $(x + 1)^2 = x^2 + 1$

```
(%i5) expand(%);
```

(%o5) $$x^2 + 2x + 1 = x^2 + 1$$

Sea A un conjunto. La aplicación identidad en A es la aplicación $1_A : A \to A$ definida como $1_A(a) = a$ para todo $a \in A$.

- Una aplicación $f : A \to B$ es biyectiva si y sólo si existe una única aplicación $g : B \to A$ tal que $g \circ f = 1_A$ y $f \circ g = 1_B$. Dicha aplicación diremos que es la inversa de f y la denotaremos por f^{-1}.

Ejercicio 14: Demuestra que la aplicación $f : \mathbb{Q} \to \mathbb{Q}$, $f(x) = \frac{1}{3}(2x + 1)$ es biyectiva. Calcula f^{-1}.

`Maxima 7`: Veamos que la inversa de la función $f(x) = x+1$ (suponemos que el dominio y codominio son los números enteros) es $g(x) = x - 1$.

```
(%i1) f(x):=x+1$ g(x):=x-1$
(%i3) f(g(x)); g(f(x));
```

(%o3) $$x$$

(%o4) $$x$$

`Maxima 8`: Consideremos ahora la aplicación $f : \{0, 1, \ldots, 7\} \to \{0, 1, \ldots, 7\}$, que dado un elemento x de $\{0, 1, \ldots, 7\}$, devuelve el resto de dividir por 8 la cantidad $x^2 + 1$.

```
(%i1) s:setify(makelist(i,i,0,7));
```

(%o1) $\{0, 1, 2, 3, 4, 5, 6, 7\}$

```
(%i2) f(x):=remainder(x^2+1,8)$
```

Calculemos el conjunto imagen de f.

```
(%i3) makelist(f(x),x,0,7);
```

(%o3) $[1, 2, 5, 2, 1, 2, 5, 2]$

```
(%i4) setify(%);
```

(%o4) $\{1, 2, 5\}$

Por lo que esta aplicación no es sobreyectiva (por ejemplo, el 0 no está en la imagen).

Veamos ahora quién es la preimagen del 1. Para ello calculamos todos los elementos que se aplican en él por f.

```
(%i5) subset(s,lambda([x],is(f(x)=1)));
```

(%o5) $\{0, 4\}$

Esto nos dice que $f(0) = f(4) = 1$, por lo que f tampoco es inyectiva.

Por último, para cualquier aplicación $f : X \to Y$ podemos definir R_f, que es una relación de equivalencia en X, de la siguiente forma

$$x \; R_f \; y \text{ si } f(x) = f(y).$$

Veamos el conjunto de clases de equivalencia en nuestro ejemplo bajo esta relación.

```
(%i6) equiv_classes(s,lambda([x,y],is(f(x)=f(y))));
```

(%o6) $\{\{0, 4\}, \{1, 3, 5, 7\}, \{2, 6\}\}$

Índice alfabético